



# Modelica.Fluid.Valvesについて (3)

## 気体用バルブ ValveCompressible



$\dot{m}$



$$\sqrt{\frac{\Delta p}{p_1}}$$

第14回 Modelica ライブラリ勉強会

2019/09/28 finback

# 気体用バルブモデル

Modelica Standard Library (MSL) に含まれる **ValveCompressible** は気体用のバルブモデルです。気体用バルブの特徴は、圧力降下比(出入口の圧力差と入口圧力の比)が臨界圧力降下比を超えると流量が差圧に依存しなくなることです。モデル化には、**流量係数**( $A_v$ ,  $C_v$ ,  $K_v$ など)と**固有流量特性**(流量係数の開度依存性)の他に**臨界圧力降下比**( $XT$ )に関する情報が必要となります。ValveCompressible は **IEC 534/ISA S.75** の規格に基づいてモデル化されていますが、**バルブメーカーの資料に見られる流量計算式**は数種類あります。例題では、これらの流量計算式に対応する改造も試みました。

## Contents

### Modelica.Fluid.Valves のバルブモデルの種類

#### クラスの継承関係

#### 気体用バルブの特徴


#### 開度を変化させた場合の流量計算式

#### ValveCompressible の計算式


#### 代表的なバルブメーカーの無次元関数

#### 例題 ValveExample3


#### まとめ



# Modelica.Fluid.Valves のバルブモデルの種類




**ValveLinear**  
流量と開度・圧力差が比例する




**ValveDiscrete**  
流量と圧力差が比例する  
全開と全閉のみ

簡易バルブモデル




**ValveIncompressible**  
液体用バルブ




**ValveVaporizing**  
内部で液体が蒸発するバルブ


バルブ選定用の  
性能データを用いたモデル




**ValveCompressible**  
気体用バルブ




**PartialTwoPortTransport**  
内部に流体を蓄えないモデル




**ValveIncompressible**




液体用バルブ




**ValveVaporizing**



内部で液体  
が蒸発する  
バルブ



**ValveCompressible**



気体用バルブ



# 気体用バルブの特徴

## 全開時の特徴

### 圧力降下比 (pressure drop ratio)

$$x \equiv \frac{\Delta p}{p_1}$$

$$\Delta p = p_1 - p_2$$

バルブのふるまいは、  
3つの領域に分けられる。

(1) 圧力降下比が小さいとき ( $x < 0.02$ )

(2) 圧力降下比が大きいとき ( $x \geq x_T$ )  $\Rightarrow \frac{\dot{m}\sqrt{RT_1}}{p_1}$  が一定になる

(3) 圧力降下比が中間のとき ( $0 < x < x_T$ )

$\dot{m}$  = 質量流量 [kg/s]


$p_1$  = 上流圧力 [Pa]

$T_1$  = 上流温度 [K]

$p_2$  = 下流圧力 [Pa]

$R$  = 気体定数 [J/kg.K]

$A_v$  = 全開時の流量係数 [m<sup>2</sup>]



$x_T$  臨界圧力降下比

## 気体用バルブ 気体用バルブの特徴

(1)  $x$  が小さいとき ( $x < 0.02$ )

液体用バルブの流量計算式を変形して近似的な関係式を導く

$$\boxed{\dot{m} = A_v \sqrt{\rho_1 \Delta p}} = A_v \sqrt{\frac{p_1}{RT_1} \Delta p} = \frac{p_1 A_v}{\sqrt{RT_1}} \sqrt{\frac{\Delta p}{p_1}} = \frac{p_1}{\sqrt{RT_1}} A_v \sqrt{x}$$


液体用バルブの流量計算式

理想気体の状態方程式

$$\rho_1 = \frac{p_1}{RT_1} \text{ を代入する。}$$


$$\Leftrightarrow \frac{\dot{m} \sqrt{RT_1}}{p_1} = A_v \sqrt{x} \quad [\text{m}^2]$$

縦軸      勾配      横軸



**最縮流部 (vena contracta) でチヨーク流れ (choked flow) となる**

## 流量から圧力差は求められない



(3)  $0 < x < x_T$  のとき

無次元関数  $Y(x)$  を導入して近似する


$$\dot{m} = \frac{p_1}{\sqrt{RT_1}} A_v \sqrt{x} Y(x)$$

と仮定する。


$Y(x)$  は無次元関数で以下を満たす。

$$Y(0) = 1$$

$$Y(x) = Y(x_T) \quad \text{for } x \geq x_T$$



無次元関数  $Y(x)$  は  
いろいろなモデル式がある。





## 開度を変化させた場合の流量計算式

$$\dot{m} = \frac{p_1}{\sqrt{RT_1}} (\text{valveCharacteristic}(\text{opening}) A_v) \sqrt{x_s} Y(x_s)$$


$$= (\text{valveCharacteristic}(\text{opening}) A_v) Y(x_s) \sqrt{\rho} \sqrt{p_1 x_s}$$

$$x_s = \begin{cases} x = \frac{\Delta p}{p}, & x < x_T \\ x_T, & x \geq x_T \end{cases}$$


臨界圧力降下比がバルブ開度に依存して変化する。

$$x_T \equiv F_{xt} = F_{xt\_full} \cdot \text{xtCharacteristic}(\text{opening})$$

バルブ開度の関数 ↗



臨界圧力降下特性



# ValveCompressible の無次元関数

## ValveCompressible の方程式

equation

```
p = max(port_a.p, port_b.p);
Fxt = Fxt_full*xtCharacteristic(opening_actual)
x = dp/p;
xs = max(-Fxt, min(x, Fxt));
Y = 1 - abs(xs)/(3*Fxt);
// m_flow = valveCharacteristic(opening)*Av*Y*sqrt(d)*sqrt(p*xs);
if checkValve then
  m_flow = homotopy(valveCharacteristic(opening_actual)*Av*Y*sqrt(Medium.density(state_a))*
    (if xs>=0 then Utilities.regRoot(p*xs, dp_turbulent) else 0),
    valveCharacteristic(opening_actual)*m_flow_nominal*dp/dp_nominal);
elseif not allowFlowReversal then
  m_flow = homotopy(valveCharacteristic(opening_actual)*Av*Y*sqrt(Medium.density(state_a))*
    Utilities.regRoot(p*xs, dp_turbulent),
    valveCharacteristic(opening_actual)*m_flow_nominal*dp/dp_nominal);
else
  m_flow = homotopy(valveCharacteristic(opening_actual)*Av*Y*
    Utilities.regRoot2(p*xs, dp_turbulent, Medium.density(state_a), Medium.density(state_b)),
    valveCharacteristic(opening_actual)*m_flow_nominal*dp/dp_nominal);
/* alternative formulation using smooth(0, ...) -- should not be used as regRoot2 has continuous derivatives
-- cf. ModelicaTest.Fluid.TestPipesAndValves.DynamicPipeInitialization --
m_flow = homotopy(valveCharacteristic(opening_actual)*Av*Y*
  smooth(0, Utilities.regRoot(p*xs, dp_turbulent)*
    (if xs>=0 then sqrt(Medium.density(state_a)) else sqrt(Medium.density(state_b)))),
  valveCharacteristic(opening_actual)*m_flow_nominal*dp/dp_nominal);
*/
end if;
```

$$x_T = F_{xt} = F_{xt\_full} \cdot \text{xtCharacteristic}(\text{opening})$$

$$Y(x_s) \equiv 1 - \frac{|x_s|}{3F_{xt}} \quad \text{無次元関数}$$

$$x_s(x) = \begin{cases} -F_{xt}, & x \leq -F_{xt} \\ x = \frac{\Delta p}{p}, & -F_{xt} < x < F_{xt} \\ F_{xt}, & x \geq F_{xt} \end{cases}$$

逆止弁機能あり

逆流あり

逆流なし

$$\dot{m} = (\text{valveCharacteristic}(\text{opening}) A_v) Y(x_s) \sqrt{\rho} \sqrt{p_1 x_s}$$

# 代表的なバルブメーカーの流量計算式

- 日阪製作所
- 巴バルブ
- フジキン
- スウェージロック
- Masoneilan Products
- EMERSON (Standardized Method)
- EMERSON (Traditional Method)

日阪製作所の計算式 <http://www.hisaka.co.jp/valve/techDoc/techDoc01.html>


	$p_2 > \frac{p_1}{2}, \Delta p < \frac{p_1}{2}$	$p_2 < \frac{p_1}{2}, \Delta p > \frac{p_1}{2}$
常温	$C_v = \frac{Q_g}{24} \sqrt{\frac{G_g}{\Delta p \times p_m}}$	$C_v = \frac{Q_g}{14.7 \times p_1} \sqrt{G_g}$
高温	$C_v = \frac{Q_g}{404} \sqrt{\frac{G_g(273 + t)}{\Delta p \times p_m}}$	$C_v = \frac{Q_g}{248 \times p_1} \sqrt{G_g(273 + t)}$

$p_1$  [kgf/cm<sup>2</sup> abs]: 1次側絶対圧  
 $p_2$  [kgf/cm<sup>2</sup> abs]: 2次側絶対圧  
 $Q_g$  [Nm<sup>3</sup>/h]: 標準状態における流量  
 $t$  [degC]: 流体温度  
 $\Delta p = p_1 - p_2$  : 差圧  
 $p_m = \frac{p_1 + p_2}{2}$  : 平均圧力

高温の式は、 $t \sim 10$ [degC]とすると常温の式と一致する。  
 高温の式を  $p_2 = p_1 - \Delta p$  を代入して変形する。

$$Q_g = \frac{404C_v p_1}{\sqrt{G_g(273 + t)}} \sqrt{\frac{\Delta p}{p_1}} \sqrt{1 - \frac{1}{2} \frac{\Delta p}{p_1}}, \quad \frac{\Delta p}{p} < \frac{1}{2}$$

$$Q_g = \frac{404.98C_v p_1}{\sqrt{G_g(273 + t)}} \sqrt{\frac{1}{2}} \sqrt{1 - \left(\frac{1}{2}\right)^2}, \quad \frac{\Delta p}{p} > \frac{1}{2}$$



$$A_v = 24.02 \times 10^{-6} C_v$$

$$Y(x) = \sqrt{1 - \frac{x}{2}}$$

$$x_T = \frac{1}{2}$$

巴バルブの計算式 <http://www.tomoevalve.com/pdf/technical.pdf>

亜臨界	$\Delta p < F_L^2 \frac{p_1}{2}$	$C_v = \frac{Q}{3.01} \sqrt{\frac{GT}{\Delta p(p_1 + p_2)}}$
臨界	$\Delta p \geq F_L^2 \frac{p_1}{2}$	$C_v = \frac{Q\sqrt{GT}}{2.62F_L \cdot p_1}$

$P_1$  : 弁入口側圧力 (kPaA)

$P_2$  : 弁出口側圧力 (kPaA)

$\Delta p = p_1 - p_2$  : 弁前後の差圧 (kPa)

$F_L$  : 圧力回復係数

$Q$  : 容積流量[15°C, 1atom](m<sup>3</sup>/h)


$G$  : 空気[15°C]に対する比重

$T$  : 流体温度(K)

$$\frac{\Delta p}{p_1} < \frac{1}{2} F_L^2 \quad Q = \frac{3.01 \times \sqrt{2} C_v p_1}{\sqrt{GT}} \sqrt{\frac{\Delta p}{p_1}} \sqrt{1 - \frac{1}{2} \frac{\Delta p}{p_1}}$$

$$\frac{\Delta p}{p_1} = \frac{1}{2} F_L^2 \quad Q = \frac{3.01 \times \sqrt{2} C_v p_1}{\sqrt{GT}} \sqrt{\frac{F_L^2}{2}} \sqrt{1 - \frac{F_L^2}{4}}$$

$$\frac{\Delta p}{p_1} \geq \frac{1}{2} F_L^2 \quad Q = \frac{2.62 C_v p_1}{\sqrt{GT}} F_L$$



$$A_v = 24.02 \times 10^{-6} C_v$$

$$Y(x) = \sqrt{1 - \frac{x}{2}}$$

$$x_T = \frac{1}{2} F_L^2 \Leftrightarrow F_L = \sqrt{2x_T}$$

$$F_L^2 = 0.985 \quad \text{のとき} \quad 3.01 \sqrt{1 - \frac{F_L^2}{4}} = 2.62$$

となり亜臨界の式と臨界の式が近似的に連続となる。

## フジキンの計算式 <https://www.fujikin.co.jp/support/calculator/>

$p_2 > \frac{p_1}{2}$	$p_2 \leq \frac{p_1}{2}$
$Q_g = \frac{4140 \cdot C_v}{\sqrt{\frac{G_G(273+t)}{(P_1 - P_2)P_2}}}$	$Q_g = \frac{2070 \cdot p_1 \cdot C_v}{\sqrt{G_G(273+t)}}$

$P_1$  [MPa abs]: 1次側絶対圧力

$P_2$  [MPa abs]: 2次側絶対圧力

$Q_g$  [m<sup>3</sup>/h (normal)]: (15℃ 0.1013MPa abs)における気体の流量


$t$  [℃]: 流体温度

$G_G$  [-]: 気体の比重(空気=1とした場合)

$p_2 = p_1 - \Delta p$  を代入して変形する。

$$Q_g = \frac{4140 \cdot C_v \cdot p_1}{\sqrt{G_G(273+t)}} \cdot \sqrt{\frac{\Delta p}{p_1}} \sqrt{1 - \frac{\Delta p}{p_1}}, \quad p_2 > \frac{p_1}{2}$$

$$Q_g = \frac{4140 \cdot C_v \cdot p_1}{\sqrt{G_G(273+t)}} \cdot \sqrt{\frac{1}{2}} \sqrt{1 - \frac{1}{2}}, \quad p_2 \leq \frac{p_1}{2}$$



$$A_v = 24.02 \times 10^{-6} C_v$$

$$Y(x) = \sqrt{1 - x}$$

$$x_T = \frac{1}{2}$$



## スウェーヂロックの計算式

<https://www.swagelok.co.jp/downloads/webcatalogs/jp/MS-06-84.pdf>

$$q = N_2 C_v p_1 \left( 1 - \frac{2\Delta p}{3p_1} \right) \sqrt{\frac{\Delta p}{p_1 G_g T_1}} \quad p_2 > \frac{1}{2} p_1$$

$$q = 0.471 N_2 C_v p_1 \sqrt{\frac{1}{G_g T_1}} \quad p_2 < \frac{1}{2} p_1$$


$q$  = 流量  
 $N_2$  = 定数 (単位によってことなる)  
 $p_1$  = 一次側圧力  
 $p_2$  = 二次側圧力  
 $\Delta p$  = 差圧 ( $p_1 - p_2$ )  
 $G_g$  = 気体の比重 (空気=1.0)  
 $T_1$  = 一次側の絶対温度

std L/min  
 6950  
 bar  
 bar  
 bar  
 bar  
 K

$$q = \frac{N_2 C_v p_1}{\sqrt{G_g T_1}} \sqrt{\frac{\Delta p}{p_1}} \left( 1 - \frac{2}{3} \frac{\Delta p}{p_1} \right)$$

$$q = \frac{N_2 C_v p_1}{\sqrt{G_g T_1}} \sqrt{\frac{1}{2}} \left( 1 - \frac{2}{3} \times \frac{1}{2} \right)$$

0.471



$$A_v = 24.02 \times 10^{-6} C_v$$

$$Y(x) = 1 - \frac{2}{3} x$$

$$x_T = \frac{1}{2}$$

## Masoneilan Products の計算式 (コントロールバルブサイジングハンドブック - AWS)

Masoneilan Products コントロールバルブサイジングハンドブック, 日本ドレッサー株式会社

$$C_v = \frac{W}{N_8 \cdot F_p \cdot P_1 \cdot Y} \sqrt{\frac{T_1 \cdot Z}{x \cdot M}}, \quad Y = 1 - \frac{x}{3F_\gamma \cdot x_{TP}}, \quad x < F_\gamma \cdot x_{TP} \quad (\text{亜臨界条件の圧縮性流体})$$

$$C_v = \frac{W}{0.667N_8 \cdot F_p \cdot P_1} \sqrt{\frac{T_1 \cdot Z}{F_\gamma \cdot x_{TP} \cdot M}}, \quad x \geq F_\gamma \cdot x_{TP} \quad (\text{臨界条件の圧縮性流体})$$

$W$  : 質量流量

$p_1$  : 弁入口絶対圧力

$p_2$  : 弁出口絶対圧力

$\Delta p = p_1 - p_2$  : バブル差圧

$x = \Delta p / p_1$  : 圧力降下比

$x_T$  : バルブ単体の圧力降下比係数 (チョークする圧力降下比)

$x_{TP}$  : レデューサ付きバルブの圧力降下比係数

$W = \text{kg/h}, p = \text{kPa}$  のとき  
 $N_8 = 0.948$   
 (上記資料P.10, 表2参照)

$F_\gamma = \gamma / 1.40$  : 比熱比係数 (空気=1.0)

$N_8$  : 単位による数値定数

$F_p$  : 配管形状係数(レデューサ補正係数)


$Y$  : 気体の膨張係数 理想気体の状態方程式に対する実在気体の補正係数

$M$  : 気体のモル質量

$Z$  : 気体の圧縮係数 ( $pV = ZRT$ )

$$W = \frac{N_8(F_p C_v) p_1}{\sqrt{\frac{Z}{M} T_1}} \sqrt{x Y} \quad F_p C_v: \text{レデューサ付きバルブの流量係数}$$

$$W = \frac{N_8(F_p C_v) p_1}{\sqrt{\frac{Z}{M} T_1}} \sqrt{F_\gamma x_{TP}} \cdot \frac{2}{3}$$



$$A_v = 24.02 \times 10^{-6} F_p C_v$$

$$Y(x) = 1 - \frac{x}{3x_T}$$

$$x_T = F_\gamma x_{TP}$$

**EMERSON** <https://www.emerson.com/documents/automation/application-guide-lp-gas-regulators-equipment-technical-section-en-126594.pdf>

# Valve Sizing Calculations (Standardized Method)

$$C_v = \frac{q}{N_7 F_p P_1 Y \sqrt{\frac{x}{G_g T_1 Z}}}$$

$$C_v = \frac{q}{N_7 F_p P_1 Y \sqrt{\frac{x}{M T_1 Z}}}$$

$$C_v = \frac{w}{N_6 F_p Y \sqrt{x P_1 \gamma_1}}$$

$$C_v = \frac{w}{N_8 F_p P_1 Y \sqrt{\frac{x M}{T_1 Z}}}$$

$$Y = 1 - \frac{x}{3 F_k x_{TP}}$$

$F_k = k/1.4$  ratio of specific heats factor  
 $x_{TP}$  pressure drop ratio factor (with piping geometry)

$q$  [Nm<sup>3</sup>/h, SCFH] Volume rate of flow

$w$  [kg/hr, pound/hr] Mass rate of flow

$N$  Numerical constant

$F_p$  [-] Piping geometry factor

$p_1$  [kPa, bar, psia] Upstream static pressure

$Y$  [-] Expansion factor

$x = \Delta p/p_1$  Ratio of pressure drop

$T_1$  [K, R (Rankine)] Absolute upstream Temperature

$Z$  [-] Compressibility factor

$G_g$  [-] specific gravity


$M$  [-] Molecular weight

$\gamma_1$  [kg/m<sup>3</sup>, pound/ft<sup>3</sup>] Specific weight

extracted from ANSI/ISA S75.01  
 IEC Standard 543-2-1, 534-2-2

$$q = \frac{N_7 (F_p C_v) P_1}{\sqrt{G_g T_1 Z}} \sqrt{x} Y$$

$$N_7 = 1360 \quad (\text{psia, SCFH})$$



$$A_v = 24.02 \times 10^{-6} (F_p C_v)$$

$$Y(x) = 1 - \frac{x}{3 x_T}$$

$$x_T = F_k x_{TP}$$


**EMERSON** <https://www.emerson.com/documents/automation/application-guide-lp-gas-regulators-equipment-technical-section-en-126594.pdf> p.330

## Valve Sizing Calculations (Traditional Method)


$$\begin{aligned} \textcircled{1} \quad Q_{SCFH} &= 59.64 C_v p_1 \sqrt{\frac{\Delta p}{p_1}} \sqrt{\frac{520}{GT}} & \frac{\Delta p}{p_1} < 0.02 \\ \textcircled{2} \quad Q_{critical} &= C_g p_1 \sqrt{\frac{520}{GT}} & \frac{59.64}{C_1} \sqrt{\frac{\Delta p}{p_1}} \geq \frac{\pi}{2} \\ \textcircled{3} \quad Q_{SCFH} &= \sqrt{\frac{520}{GT}} C_g p_1 \sin \left[ \frac{59.64}{C_1} \sqrt{\frac{\Delta p}{p_1}} \right] & 0 < \frac{59.64}{C_1} \sqrt{\frac{\Delta p}{p_1}} < \frac{\pi}{2} \end{aligned}$$

$Q_{SCFH}$  = gas flow rate,  
 SCFH(cubic feet per hour at 60 °F air)  
 $p_1$  = valve inlet pressure, psi  
 $\Delta p$  = pressure drop, psi  
 $G$  = specific gravity at 60 °F (520 degrees Rankine )  
 $Q_{critical}$  = critical flow rate, SCFH  
 $C_1 = C_g / C_v$   
 $C_g$  = gas sizing coefficient

## $C_1$ Gas Flow Factor




## SI単位



③  $0 < x < x_T$  のとき、

$$\begin{aligned}\frac{\dot{m}\sqrt{RT_1}}{p_1} &= \frac{A_v C_1}{59.64} \sin\left(\frac{59.64}{C_1}\sqrt{x}\right) \\ &= A_v \sqrt{x} \frac{\sin\left(\frac{59.64}{C_1}\sqrt{x}\right)}{\frac{59.64}{C_1}\sqrt{x}} \\ &= A_v \sqrt{x} Y(x) \\ \frac{59.64}{C_1}\sqrt{x_T} &= \frac{\pi}{2} \Leftrightarrow x_T = \left(\frac{\pi}{2} \cdot \frac{C_1}{59.64}\right)^2\end{aligned}$$



$$\begin{aligned}A_v &= 24.02 \times 10^{-6} C_v \\ Y(x) &= \frac{\sin\left(\frac{59.64}{C_1}\sqrt{x}\right)}{\frac{59.64}{C_1}\sqrt{x}} \\ x_T &= \left(\frac{\pi}{2} \cdot \frac{C_1}{59.64}\right)^2 \\ \Leftrightarrow C_1 &= 37.97\sqrt{x_T}\end{aligned}$$

①  $x \rightarrow 0$  のとき、

$$\begin{aligned}\theta &= \frac{59.64}{C_1}\sqrt{x} \text{ とすると、} \\ \lim_{x \rightarrow 0} Y(x) &= \lim_{\theta \rightarrow 0} \frac{\sin \theta}{\theta} = \lim_{\theta \rightarrow 0} \frac{\theta - \frac{\theta^3}{3!} + \frac{\theta^5}{5!} - \dots}{\theta} = 1 \Rightarrow \frac{\dot{m}\sqrt{RT_1}}{p_1} = A_v \sqrt{x}\end{aligned}$$




②  $x \geq x_T$  のとき(臨界時)、

$$Y(x_T) = \frac{\sin\left(\frac{59.64}{C_1}\sqrt{x_T}\right)}{\frac{59.64}{C_1}\sqrt{x_T}} = \frac{\sin \frac{\pi}{2}}{\frac{\pi}{2}} = \frac{2}{\pi}$$

$$\frac{\dot{m}\sqrt{RT_1}}{p_1} = A_v \sqrt{x_T} Y(x_T) = \frac{A_v C_1}{59.64} = 1.6767 \times 10^{-2} A_v C_1$$

# 例題 ValveExample3

- AngleValveTest1 カタログを参考にしてアングルバルブをモデル化する。圧力差を固定して開度を変化させる。
- AngleValveTest2 開度を固定して圧力差を変化させる。
- AngleValveTest3 ValveCompressible を改造して、各バルブメーカーのモデルを作る。
- AngleValveTest4 ValveCompressible を改造して、CombiTable1Ds でバルブ特性が設定できるモデル ValveCompressibleE を作成する。
- AngleValveTest5 ValveCompressibleE のテストモデル。





# AngleValveTest1

参考 Fisher 461, アングルバルブ

<https://www.emerson.com/documents/automation/product-bulletin-fisher-461-increased-outlet-angle-sweep-flo-valve-en-125796.pdf>

<https://www.emerson.com/documents/automation/product-data-sheet-catalog-12-section-1-sept-2017-fisher-en-122394.pdf>

## 全開時バルブデータ (カタログより)

ポート径	d	12.7	mm
流量係数	Cv	11.7	
臨界圧力降下比	XT	0.187	

## 全開時の差圧 1.5 kPa の流量

断面積	A	0.00012668	m2
流量係数	Av	0.00028103	m2
空気密度	$\rho$	1.293	kg/m3
空気比熱比係数	Fk	1	
入口圧力	p1 = p_nominal	102825	Pa
出口圧力	p2	101325	Pa
圧力差	dp_nominal	1500	Pa
圧力降下比	x	0.01458789	
臨界圧力降下比	XT	0.187	
無次元関数	Y	0.97399663	
質量流量	m_flow_nomina	0.01205483	kg/s

流量係数 (Cv) と臨界圧力降下比 (XT) から ValveCompressible を設定する。

アングルバルブは、高粘性流体やフラッシングなどのある流体に使用されることが多い。  
臨界圧力比特性の記述があるので気体で適用してみる。

$$A = \frac{\pi D^2}{4}$$

$$A_v = 24.02 \times 10^{-6} C_v$$

$$F_k = \frac{\gamma}{\gamma_{air}} = \frac{\gamma}{1.4}, \quad \gamma = \frac{c_p}{c_v}$$

$$\Delta p = p_1 - p_2,$$

$$X_T = F_{xt} = F_k x_{TP} = F_k x_{T\_air}$$

$$Y = 1 - \frac{x}{3F_{xt}}$$

$$\dot{m} = A_v Y(x) \sqrt{\rho} \sqrt{p_1 x}$$

比熱比  
流体の種類  
によって変わる



# 流量特性と臨界圧力降下比特性（カタログより）

	Cv	XT
0	0	0.397
10	0.597	0.301
20	0.9822	0.205
30	1.38	0.186
40	1.87	0.198
50	2.54	0.206
60	3.45	0.212
70	4.91	0.196
80	7.22	0.159
90	9.95	0.16
100	11.7	0.187

外挿値

正規化する

opening	valveCharacteristic	xtCharacteristic
0	0	1
0.1	0.051025641	0.758186398
0.2	0.083948718	0.516372796
0.3	0.117948718	0.468513854
0.4	0.15982906	0.498740554
0.5	0.217094017	0.518891688
0.6	0.294871795	0.534005038
0.7	0.41965812	0.493702771
0.8	0.617094017	0.400503778
0.9	0.85042735	0.40302267
1	1	0.471032746



Excelで求めた近似曲線を設定した関数を作成する。

## valveCharacteristic 用の関数の作成

```
function avCurve1
  extends Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun;
algorithm
  rc := (((((-9.1683 * pos) + 19.855) * pos - 12.964) * pos + 3.0747) * pos + 0.2061) * pos;
end avCurve1;
```

## xtCharacteristic 用の関数の作成


```
function xtCurve1
  extends Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun;
algorithm
  rc := ((((((((-45.881 * pos) + 152.17) * pos - 183.4) * pos + 94.103) * pos - 15.827) * pos - 1.6945) * pos + 1;
end xtCurve1;
```

## テストモデルの作成

下流圧力大気圧、上流と下流の差圧 1.5 kPa とし、  
開度を全閉から全開まで変化させる。

**Modelica.Blocks.Sources.Ramp**

height = 1  
duration = 10 [s]  
offset = 0  
startTime = 0 [s]



**Boundary\_pT**

$p = 102825$  [Pa]  
 $T = 273.15$  [K]

**ValveCompressible**

$CvData = Modelica.Fluid.Types.CvTypes.Cv$   
 $Cv = 11.7$   
 $Fxt\_full = 0.397$   
 $p\_nominal = 102825$  [Pa]  
 $dp\_nominal = 0.015$  [bar]  
 $m\_flow\_nominal = 0.01205483$  [kg/s]

**Boundary\_pT**

$p = 101325$  [Pa]  
 $T = 273.15$  [K]

## ソースコードの編集

```


model AngleValveTest1
  replaceable package Medium = Modelica.Media.Air.DryAirNasa;
  Modelica.Fluid.Sources.Boundary_pT boundary(redeclare package Medium = Medium, T = 273.15, nPorts = 1, p = 102825)
    annotation( ...);
  Modelica.Fluid.Sources.Boundary_pT boundary_pT(redeclare package Medium = Medium, T = 273.15, nPorts = 1, p = 101325)
    annotation( ...);
  Modelica.Fluid.Valves.ValveCompressible valveCompressible(
    redeclare package Medium = Medium,
    redeclare function valveCharacteristic = avCurve1,
    redeclare function xtCharacteristic = xtCurve1,
    Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv, Fxt_full = 0.397,
    dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation( ...);
  Modelica.Blocks.Sources.Ramp ramp(duration = 10, height = 1, offset = 0) annotation( ...);
  inner Modelica.Fluid.System system annotation( ...);
equation
  connect(valveCompressible.port_a, boundary.ports[1]) annotation( ...);
  connect(valveCompressible.port_b, boundary_pT.ports[1]) annotation( ...);
  connect(ramp.y, valveCompressible.opening) annotation( ...);
  annotation( ...);
end AngleValveTest1;

```

← 作成した近似式の関数を設定する。



# シミュレーション結果






# AngleValveTest2

出口圧力を下げて臨界流量になるか調べる

AngleValveTest1 と同じバルブの設定で、開度を固定して下流の圧力を変化させる。

Modelica.Blocks.Sources.Constant




Modelica.Blocks.Sources.Ramp





height = -80000  
duration = 10 [s]  
offset = 201325  
startTime = 0 [s]

Modelica.Blocks.Sources.RealExpressio

$\sqrt{\frac{\Delta p}{p_1}}$   
を計算する



## シミュレーション結果





# AngleValveTest3

メーカーの流量計算式に合わせて  
ValveCompressible の改造する。

テストモデル AngleValveTest3

圧力回復特性とGas Flow Factor

- ValveCompressible Masoneilan Products,  
EMERSON (Standized Method)など
- ValveCompressibleA 日阪製作所、巴バルブなど
- ValveCompressibleB フジキンなど
- ValveCompressibleC スウェージロックなど
- ValveCompressibleD EMERSON (Traditional Method) など


シミュレーション結果

- 質量流量のシミュレーション結果
- 臨界流量のシミュレーション結果の検証
- 各モデル比較

# テストモデル AngleValveTest3

$$y = \sqrt{\frac{\Delta p}{p_1}}$$

$$y = \text{sqrt}(\text{valveCompressible.dp} / \text{valveCompressible.p})$$



## テストモデル用の圧力回復特性と Gas Flow Factor

カタログデータには圧力回復特性と Gas Flow Factor は含まれていないので臨界圧力降下比から推定する。


$F_L$  や  $C_1$  は以下の式により  
右表のように近似する。

### 圧力回復係数

$$F_L = \sqrt{2x_T}$$

### Gas Flow Factor


$$C_1 = 37.97\sqrt{x_T}$$



	XT	FL	C1
0	0.397	0.89106678	23.9241133
10	0.301	0.77588659	20.83165838
20	0.205	0.64031242	17.19164868
30	0.186	0.60991803	16.37559426
40	0.198	0.62928531	16.89558339
50	0.206	0.64187226	17.23352852
60	0.212	0.65115282	17.4827009
70	0.196	0.62609903	16.81003559
80	0.159	0.56391489	15.14046311
90	0.16	0.56568542	15.188
100	0.187	0.61155539	16.41955567

$F_L$  または  $C_1$  のデータを正規化する。

	flCharacteristic
0	1
0.1	0.870738995
0.2	0.71859084
0.3	0.684480718
0.4	0.706215657
0.5	0.720341369
0.6	0.730756483
0.7	0.702639859
0.8	0.632853678
0.9	0.634840665
1	0.686318254




Excel で近似曲線を求める。

近似曲線の関数を作る。

**flCurve1(opening)**

```
function flCurve1
  extends Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun;
algorithm
  rc := (((((-31.992 * pos) + 107.73) * pos - 132.48) * pos + 70.548) * pos - 13.555) * pos - 0.5709) * pos + 1;
end flCurve1;
```




$$F_L = 0.89106678 \text{ flcurve1(opening)}$$

$$C_1 = 23.9241133 \text{ flcurve1(opening)}$$



## AngleValveTest3 の valveCompressible の設定

```
Fxt_full = 0.397  
CvData = Modelica.Fluid.Types.Cv  
Cv = 11.7  
p_nominal = 102825  
dp_nominal = 0.015  
m_flow_nominal = 0.01205483
```



AngleValveTest1と  
同様の設定!

### valveCompressible のコード

```
Modelica.Fluid.Valves.ValveCompressible valveCompressible(  
  redeclare package Medium = Medium,  
  redeclare function valveCharacteristic = avCurve1,  
  redeclare function xtCharacteristic = xtCurve1,  
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv, Fxt_full = 0.397,  
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation(...);
```

# ValveCompressibleA

ValveCompressible を複製して作成し編集する。

```
model ValveCompressibleA
  "Valve for compressible fluids, accounts for choked flow conditions"
  extends Modelica.Fluid.Valves.BaseClasses.PartialValve;
  import Modelica.Fluid.Types.CvTypes;
  import Modelica.Constants.pi;
  import SI = Modelica.SIunits;
  import Modelica.Fluid.Utilities;
  parameter Medium.AbsolutePressure p_nominal "Nominal inlet pressure"
  annotation(Dialog(group="Nominal operating point"));
  // parameter Real Fxt_full=0.5 "Fk*xt critical ratio at full opening";
  parameter Real FL_max = 1.0;
  //replaceable function xtCharacteristic =
  //  Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.one
  //  constrainedby
  //  Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun
  //  "Critical ratio characteristic";
  replaceable function flCharacteristic =
    Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.one
    constrainedby
    Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun
    "Pressure Recovery Factor";
  Real FL;
  Real Fxt;
  Real x "Pressure drop ratio";
  ...
```

← 圧力回復係数の最大値

← 圧力回復係数を正規化した関数

← 圧力回復係数

## ValveCompressibleA (つづき)

initial equation

```
if CvData == CvTypes.OpPoint then
```

```
// Determination of Av by the nominal operating point conditions
```

```
Fxt_nominal = 0.5 * (Fl_max*flCharacteristic(opening_nominal))^2;
```

```
...
```

$$x_T = F_{xt} = \frac{F_L^2}{2}$$

## ValveCompressibleA (つづき)

equation

```
p = max(port_a.p, port_b.p);
```

```
Fl = Fl_max * flCharacteristic(opening_actual);
```

```
Fxt = 0.5*Fl*Fl;
```

```
x = dp/p;
```

```
xs = max(-Fxt, min(x, Fxt));
```

```
Y = sqrt(1-abs(xs)/2);
```

```
...
```

$$x_T = F_{xt} = \frac{F_L^2}{2}$$

$$Y(x) = \sqrt{1 - \frac{x}{2}}$$

$$x_T = \frac{1}{2}$$


日阪製作所など  
(デフォルト)

$$x_T = \frac{F_L^2}{2}$$

巴バルブなど

## AngleValveTest3 の valveCompressibleA の設定

```
CvData = Modelica.Fluid.Types.Cv  
Cv = 11.7  
p_nominal = 102825  
dp_nominal = 0.015  
m_flow_nominal = 0.01205483
```



$$x_T = \frac{1}{2}$$

(デフォルトの設定)

### valveCompressibleA のコード

```
Modelica.Fluid.Valves.ValveCompressible valveCompressibleA(  
  redeclare package Medium = Medium,  
  redeclare function valveCharacteristic = avCurve1,  
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv,  
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation(...);
```

## AngleValveTest3 の valveCompressibleA1 の設定

**Fl\_max = 0.89106678**


CvData = Modelica.Fluid.Types.Cv

Cv = 11.7

p\_nominal = 102825

dp\_nominal = 0.015

m\_flow\_nominal = 0.01205483



圧力回復係数を設定する

$$x_T = \frac{1}{2} F_L^2$$

$$F_L = F_{L\_max} \cdot \text{flCharacteristic}(\text{opening})$$

## valveCompressibleA1のコード

```
Modelica.Fluid.Valves.ValveCompressible valveCompressibleA1(
  redeclare package Medium = Medium,
  redeclare function valveCharacteristic = avCurve1,
  redeclare function flCharacteristic = flCurve1,
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv, Fl_max = 0.89100678,
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation(...);
```

## ValveCompressibleB

同様にValveCompressible を複製して作成し編集する。

```
model ValveCompressibleB
  "Valve for compressible fluids, accounts for choked flow conditions"
  extends Modelica.Fluid.Valves.BaseClasses.PartialValve;
  import SI = Modelica.SIunits;
  import Modelica.Fluid.Utilities;
  ...
```

```
initial equation
  if CvData == CvTypes.OpPoint then
    // Determination of Av by the nominal operating point conditions
    Fxt_nominal = 0.5;
    x_nominal = dp_nominal/p_nominal;
    ...
```

$$F_{xt} = \frac{1}{2}$$

```
equation
  p = max(port_a.p, port_b.p);
  Fxt = 0.5;
  x = dp/p;
  xs = max(-Fxt, min(x, Fxt));
  Y = sqrt(1 - xs);
  ...
```


$$F_{xt} = \frac{1}{2}$$

$$Y(x) = \sqrt{1 - x} \quad \text{フジキンなど}$$



## AngleValveTest3 の valveCompressibleB の設定

```
CvData = Modelica.Fluid.Types.Cv  
Cv = 11.7  
p_nominal = 102825  
dp_nominal = 0.015  
m_flow_nominal = 0.01205483
```



### valveCompressibleBのコード

```
Modelica.Fluid.Valves.ValveCompressible valveCompressibleB(  
  redeclare package Medium = Medium,  
  redeclare function valveCharacteristic = avCurve1,  
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv,  
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation(...);
```

## ValveCompressibleC

同様にValveCompressible を複製して編集する。

```
model ValveCompressibleC
  "Valve for compressible fluids, accounts for choked flow conditions"
  extends Modelica.Fluid.Valves.BaseClasses.PartialValve;
  import SI = Modelica.SIunits;
  import Modelica.Fluid.Utilities;
  ...
```

```
initial equation
  if CvData == CvTypes.OpPoint then
    // Determination of Av by the nominal operating point conditions
    Fxt_nominal = 0.5;
    x_nominal = dp_nominal/p_nominal;
    ...
```

$$F_{xt} = \frac{1}{2}$$

```
equation
  p = max(port_a.p, port_b.p);
  Fxt = 0.5;
  x = dp/p;
  xs = max(-Fxt, min(x, Fxt));
  Y = 1 - (2/3)*abs(xs);
  ...
```


$$F_{xt} = \frac{1}{2}$$

$$Y(x) = 1 - \frac{2}{3}x$$

スウェーヂロックなど

## AngleValveTest3 の valveCompressibleC の設定

```
CvData = Modelica.Fluid.Types.Cv  
Cv = 11.7  
p_nominal = 102825  
dp_nominal = 0.015  
m_flow_nominal = 0.01205483
```



### valveCompressibleCのコード

```
Modelica.Fluid.Valves.ValveCompressible valveCompressibleC(  
  redeclare package Medium = Medium,  
  redeclare function valveCharacteristic = avCurve1,  
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv,  
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation(...);
```

## ValveCompressibleD

同様にValveCompressible を複製して作成し編集する。

```
model ValveCompressibleD "Valve for compressible fluids, accounts for choked flow conditions"
  extends Modelica.Fluid.Valves.BaseClasses.PartialValve;
  import SI = Modelica.SIunits;
  import Modelica.Fluid.Utilities;
  import Modelica.Fluid.Types.CvTypes;
  import Modelica.Constants.pi;
  parameter Medium.AbsolutePressure p_nominal "Nominal inlet pressure" annotation(
    Dialog(group = "Nominal operating point"));
  // parameter Real Fxt_full = 0.5 "Fk*xt critical ratio at full opening";
  // replaceable function xtCharacteristic = Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.one
  constrainedby
  // Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun "Critical ratio characteristic";
  parameter Real C1_full = 23.924 "Fk*xt critical ratio at full opening";
  replaceable function c1Characteristic = Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.one
  constrainedby
    Modelica.Fluid.Valves.BaseClasses.ValveCharacteristics.baseFun "Critical ratio characteristic";
  Real C1;
  Real C2;
  Real Fxt;
  ...
```

$C_1$  の基準値  
 $C_1$  を正規化した関数

## ValveCompressibleD (つづき)

initial equation

```
if CvData == CvTypes.OpPoint then
```

```
// Determination of Av by the nominal operating point conditions
```

```
Fxt_nominal = (pi/2*C1_full*c1Characteristic(opening_nominal)/59.64)^2;
```

```
x_nominal = dp_nominal / p_nominal;
```

```
...
```

$$x_T = F_{xt} = \left( \frac{\pi}{2} \cdot \frac{C_1}{59.64} \right)^2$$

## ValveCompressibleD (つづき)

equation

```
p = max(port_a.p, port_b.p);
```

```
C1 = C1_full*c1Characteristic(opening_actual);
```

```
Fxt = (0.5*pi*C1/59.64)^2;
```

```
// Fxt = Fxt_full*xtCharacteristic(opening_actual);
```

```
x = dp / p;
```

```
xs = max(-Fxt, min(x, Fxt));
```

```
C2 = 59.64*sqrt(abs(xs))/C1;
```

```
Y = sin(C2)/C2;
```


```
...
```

$$x_T = F_{xt} = \left( \frac{\pi}{2} \cdot \frac{C_1}{59.64} \right)^2$$

$$Y(x) = \frac{\sin \left( \frac{59.64}{C_1} \sqrt{x} \right)}{\frac{59.64}{C_1} \sqrt{x}}$$

## AngleValveTest3 の valveCompressibleD の設定

```
C1_full = 23.924
CvData = Modelica.Fluid.Types.Cv
Cv = 11.7
p_nominal = 102825
dp_nominal = 0.015
m_flow_nominal = 0.01205483
```



$$x_T = F_{xt} = \left( \frac{\pi}{2} \cdot \frac{C_1}{59.64} \right)^2$$




$$C_1 = C_{1\_full} \cdot c1Characteristic(opening)$$

### valveCompressibleDのコード

```
Modelica.Fluid.Valves.ValveCompressible valveCompressibleD(
  redeclare package Medium = Medium,
  redeclare function valveCharacteristic = avCurve1,
  redeclare function c1Characteristic = flCurve1,
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv, C1_full = 23.924,
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation(...);
```



シミュレーション結果



## 臨界流量のシミュレーション結果の検証

### シミュレーション結果

コンポーネント	kg/s
valveCompressible	5.837E-02
	5.837E-02
	5.837E-02
valveCompressibleA	1.241E-01
valveCompressibleA1	8.273E-02
valveCompressibleB	1.013E-01
valveCompressibleC	9.550E-02
valveCompressibleD	5.527E-02

### カタログ等の計算式で求めた流量 (手計算)

	臨界流量		kg/s	
valveCompressible	5.825E-02	kg/s	5.825E-02	
Masoneilan Products	2.096E+02	kg/h	5.823E-02	N8=0.948 (kg/h, kPa)
EMERSON(Standized)	6.041E+03	scfh	5.802E-02	N7=1360 (psia, SCFH)
日阪製作所	3.596E+02	Nm3/h	1.291E-01	
巴バルブ	2.498E+02	m3/h	8.472E-02	
フジキン	2.951E+02	m3/h	1.001E-01	
スウェーヂロック	4.665E+03	std L/min	9.494E-02	N2=6950
EMERSON(Traditional)	5.769E+03	scfh	5.540E-02	

### 計算諸元

重力加速度	g	9.80665	m/s2
流量係数	Cv	11.7	
	Av	0.000281034	m2
臨界圧力降下比	XT	0.187	
圧力回復係数	FL	0.611555394	
Gas Flow Factor	C1	16.41955567	


計算式の数値の単位はいろいろ kg/s に変換

空気(0 °C, 101325 Pa)	ρ	1.29228	kg/m3
空気(15.8 °C, 101325 Pa)		1.221	kg/m3
空気(0°C, 201325Pa)		2.56766	kg/m3
空気モル質量	M	28.9651159	g/mol

### 上流圧力

201325	Pa
201.325	kPaA
2.01325	bar
0.201325	MPa abs
2.052943666	kgf/cm2
29.19972255	psi






# AngleValveTest4


MSLの機能でバルブ特性をフィッティングしたい

バルブ特性データのフィッティングが、MSL に含まれる CombiTable1Ds の機能でできるように改造する。



## CombiTable1Ds による valveCharateristic と xtCharacteristic のフィッティング


```
Modelica.Blocks.Tables.CombiTable1Ds combiTable1Ds(
  smoothness = Modelica.Blocks.Types.Smoothness.ContinuousDerivative,
  table = [
    0, 0, 1;
    0.1, 0.051025641, 0.758186398;
    0.2, 0.083948718, 0.516372796;
    0.3, 0.117948718, 0.468513854;
    0.4, 0.15982906, 0.498740554;
    0.5, 0.217094017, 0.518891688;
    0.6, 0.294871795, 0.534005038;
    0.7, 0.41965812, 0.493702771;
    0.8, 0.617094017, 0.400503778;
    0.9, 0.85042735, 0.40302267;
    1, 1, 0.471032746])
annotation( ...);
```





# ValveCompressibleの改造

ValveCompressibleを複製して ValveCompressibleEを作成し、CombiTable1Ds を貼り付け、名前を characteristicTable にして、opening\_actual と接続する。



## ソースコードの先頭部分

```
model ValveCompressibleE
  "Valve for compressible fluids, accounts for choked flow conditions"
  extends Modelica.Fluid.Valves.BaseClasses.PartialValve;
  import SI = Modelica.SIunits;
  import Modelica.Fluid.Utilities;
  ...
```

characteristicTable のデフォルト

```
...
SI.AbsolutePressure dp_turbulent = if not use_Re then dp_small else
    max(dp_small,
        (Medium.dynamicViscosity(state_a) + Medium.dynamicViscosity(state_b))^2*pi/8*Re_turbulent^2
        /(max(valveCharacteristic(opening_actual),0.001)
        *Av*Y*(Medium.density(state_a) + Medium.density(state_b)))
    );
Modelica.Blocks.Tables.CombiTable1Ds charactericticTable(table = [0, 0, 0; 1, 1, 1]) annotation( ...);
protected
...
```

デフォルト設定はリニアにする。

equation の characteristicTable の接続部分

```
equation
...
connect(opening_actual, charactericticTable.u) annotation( ...);
...
```



characteristicTable を参照して臨界圧力降下比特性と固有流量特性を設定する。

equation

```

p = max(port_a.p, port_b.p);
// Fxt = Fxt_full*xCharacteristic(opening_actual);
Fxt = Fxt_full*characteristicTable.y[2]; ← 臨界圧力降下比特性の設定
x = dp/p;
xs = max(-Fxt, min(x, Fxt));
Y = 1 - abs(xs)/(3*Fxt);
// m_flow = valveCharacteristic(opening)*Av*Y*sqrt(d)*sqrt(p*xs);
if checkValve then
    m_flow = homotopy(characteristicTable.y[1]*Av*Y*sqrt(Medium.density(state_a))*
        (if xs>=0 then Utilities.regRoot(p*xs, dp_turbulent) else 0),
        characteristicTable.y[1]*m_flow_nominal*dp/dp_nominal);
elseif not allowFlowReversal then
    m_flow = homotopy(characteristicTable.y[1]*Av*Y*sqrt(Medium.density(state_a))*
        Utilities.regRoot(p*xs, dp_turbulent),
        characteristicTable.y[1]*m_flow_nominal*dp/dp_nominal);
else
    m_flow = homotopy(characteristicTable.y[1]*Av*Y*
        Utilities.regRoot2(p*xs, dp_turbulent, Medium.density(state_a),
        Medium.density(state_b)),
        characteristicTable.y[1]*m_flow_nominal*dp/dp_nominal);
...

```


流量計算部分の  
固有流量特性の設定

# テストモデル AngleValveTest4




AngleValveTest1 と同様のモデルで  
バルブモデルで ValveCompressibleE と  
ValeCompressible を比較する。

## valveCompressivleEの設定

```
ValveExample3.ValveCompressibleE valveCompressibleE(
  redeclare package Medium = Medium,
  charactericticTable(
    smoothness = Modelica.Blocks.Types.Smoothness.ContinuousDerivative,
    table = [
      0, 0, 1;
      0.1, 0.051025641, 0.758186398;
      0.2, 0.083948718, 0.516372796;
      0.3, 0.117948718, 0.468513854;
      0.4, 0.15982906, 0.498740554;
      0.5, 0.217094017, 0.518891688;
      0.6, 0.294871795, 0.534005038;
      0.7, 0.41965812, 0.493702771;
      0.8, 0.617094017, 0.400503778;
      0.9, 0.85042735, 0.40302267;
      1, 1, 0.471032746]),
  Cv = 11.7, CvData = Modelica.Fluid.Types.CvTypes.Cv, Fxt_full = 0.397,
  dp_nominal = 1500, m_flow_nominal = 0.01205483, p_nominal = 102825) annotation( ...);
```






# シミュレーション結果





# テストモデル AngleValveTest5

AngleValveTest2 と同様のモデルで  
バルブモデルを ValveCompressibleE に  
変更する。



## まとめ

- MSL に含まれる液体用のバルブモデルである ValveCompressible について、モデルの計算式や使用方法を調査した。
- ValveCompressible は、流量係数( $A_v$ ,  $K_v$ ,  $C_v$ など)、固有流量特性(valveCharacteristic)、臨界圧力降下比特性(XT, xtCharacteristic)を使用してモデル化する。
- バルブメーカー数社の流量計算式を再現するため、ValveCompressible を改造し、シミュレーション結果を比較した。

**Licensed by Amane Tanaka under the Modelica License 2**

**Copyright(c) 2019, Amane Tanaka**

- The purpose of this document is introducing the valve models of Modelica.Fluid which are included in the Modelica Standard Library (MSL). This document uses libraries, software, figures, and documents included in MSL, and those modifications. Licenses and copyrights of those are written in next page.
- This document is free and the use is completely at your own risk; it can be redistributed and/or modified under the terms of the Modelica license 2, see the license conditions (including the disclaimer of warranty) at <http://www.modelica.org/licenses/ModelicaLicense2>

# Modelica Standard Library License

<https://github.com/modelica/ModelicaStandardLibrary/blob/master/LICENSE>

## BSD 3-Clause License

Copyright (c) 1998-2019, Modelica Association and contributors  
All rights reserved.

Redistribution and use in source and binary forms, with or without modification, are permitted provided that the following conditions are met:

- \* Redistributions of source code must retain the above copyright notice, this list of conditions and the following disclaimer.
- \* Redistributions in binary form must reproduce the above copyright notice, this list of conditions and the following disclaimer in the documentation and/or other materials provided with the distribution.
- \* Neither the name of the copyright holder nor the names of its contributors may be used to endorse or promote products derived from this software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.